* ==============================================================================
*
* Date:	Summer 2024
* Paper: Climate shocks, regional favoritism and trust in leaders: Insights from droughts in Africa
*
* This creates Table 4
*
* ==============================================================================

* Open dataset
cd $PATH
use tr_rep, clear

* Global macros

* Control sets
global xcont "age age2 male i.ed_ord"

global fecont "urban ea_svc_a ea_svc_b ea_svc_c ea_fac_a ea_fac_b ea_fac_c ea_fac_d ea_fac_e ea_sec_a ea_sec_b grs0 grsem0 grsl cropland_d ln_alt ln_dist_cit ln_alt_sd ln_dist_bor ln_dist_coa ln_dist_riv ln_dist_roa i.year i.month i.uid1 countrytrend*"

* Clustering level
global clustervar "cidgrid01"


* Clean sample
keep if precision_code<=3
global zxcont "age age2 male ed_ord"
global zfecont "urban ea_svc_a ea_svc_b ea_svc_c ea_fac_a ea_fac_b ea_fac_c ea_fac_d ea_fac_e ea_sec_a ea_sec_b grs0 grsem0 grsl cropland_d ln_alt ln_dist_cit ln_alt_sd ln_dist_bor ln_dist_coa ln_dist_riv ln_dist_roa year"

foreach var of varlist $zxcont $zfecont {
	drop if `var'==.
}

* Standardize
egen z=std(aiclwmcgrwsp)
replace aiclwmcgrwsp=z
drop z

* Drought condition dummies
gen dr0=cond(aiclwmcgrwsp>0.0, 1, 0, .) if aiclwmcgrwsp!=.
label var dr0 "aiclwmcgrws>0"
gen dr0_10=cond(aiclwmcgrwsp>0.0 & aiclwmcgrwsp<=1.0, 1, 0, .) if aiclwmcgrwsp!=.
label var dr0_10 "0<aiclwmcgrwsp<=1"
gen dr10_15=cond(aiclwmcgrwsp>1.0 & aiclwmcgrwsp<=1.5, 1, 0, .) if aiclwmcgrwsp!=.
label var dr10_15 "1.0<aiclwmcgrwsp<=1.5"
gen byte dr15=cond(aiclwmcgrwsp>1.5, 1, 0, .) if aiclwmcgrwsp!=.
label var dr15 "aiclwmcgrwsp>1.5"

* Table name
global tabname "severity"

* Regression(s)
timer clear
local x=0
local z = ""

* Drought condition dummies set
global drvars "dr0_10 dr10_15 dr15"

*c
local x=`x'+1
di "Column `x'"
timer on `x'
qui reg tr_pres_ord dr0 $xcont $fecont, vce(cluster $clustervar)
local z`x' m`x'
local z = "`z'" + " `z`x''"
est store m`x'
local m="c`x'"
estimates save $tabname`m', replace
timer off `x'
timer list `x'

*c
local x=`x'+1
di "Column `x'"
timer on `x'
qui reg tr_pres_ord $drvars $xcont $fecont, vce(cluster $clustervar)
local z`x' m`x'
local z = "`z'" + " `z`x''"
est store m`x'
local m="c`x'"
estimates save $tabname`m', replace
timer off `x'
timer list `x'

*c
local x=`x'+1
di "Column `x'"
timer on `x'
qui reg tr_pres_ord dr15 $xcont $fecont if dr0==1, vce(cluster $clustervar)
local z`x' m`x'
local z = "`z'" + " `z`x''"
est store m`x'
local m="c`x'"
estimates save $tabname`m', replace
timer off `x'
timer list `x'





* Post-table 
timer list   
di "`z'"

* Simplified table generation
global keepvars "dr0 $drvars"

* In Stata output window
esttab `z' , ///
	mtitles("All" "All" "Dry Conditions") ///
	addnotes("Dependent variable: Trust in President" "Standard errors clustered on 0.1-degree grid cells") ///
	keep($keepvars) ///
	order($keepvars) ///
	cells(b(star fmt(%8.3f) ) se( par fmt(%6.3f)) ) ///
	stats(N r2, fmt(%9.0f %9.2f)) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) r2(3) ///
	legend


* Excel-file
estout `z' using $OUTPATH/$tabname.xls, replace ///
	mlabels("All" "All") ///
	postfoot("Dependent variable: Trust in President" "Standard errors clustered on 0.1-degree grid cells") ///
	keep($keepvars) ///
	order($keepvars) ///
	cells(b(star fmt(%8.3f) ) se( par(`"="("' `")""') fmt(%6.3f)) ) ///
	stats(N r2, fmt(%9.0f %9.2f)) ///
	starlevels(* 0.10 ** 0.05 *** 0.01) ///
	legend



